Exercises for PC SAS Class

Let’s try a couple of exercises to practice using SAS.  I’ll do the first one with you, and then you can try the second one your own.

We will use a data set called states.txt.  The data and a description (reproduced below) of the data can be found on the course web site:

http://www.andrew.cmu.edu/course/90-770/index.htm

The columns are 2-letter postal abbreviation of state name, then the state population in thousands of people, then the income per capita in dollars, then the state government spending in millions of dollars.

Exercise 1. 

1. Start SAS

2. Bring in the ASCII data states.txt from the course web site, create a per capita spending variable, and save the data as a permanent SAS data set called c:\users\stperm.sd2.

Remember:  You will need to create a libname and filename before the data statement.

Hint:  See pages 4 and 5 of the handout for help.

3. Display the contents of your data file.

4. Display the means of per capita income and per capita spending.

5. Display one observation at time using Proc Fsbrowse.

6. Create a new temporary data set that contains only the variables state, per capita income, and income for all states other than New York and California.

Note your output both in the Log and Output windows.

Exercise 2.

Write one SAS program to do all of the following:

1. Bring in SAS data set c:\users\stperm.sd2 into a new temporary data set.

2. Create a dummy variable that takes on the value 1 if a state’s population is greater than 5,000,000 (big population state) and equals 0 otherwise (small population state). 

Remember:  population is measured in thousands.

3. Display a frequency distribution of your dummy variable.

4. Calculate the means of spending and per capita spending for big and small states.

5. Create a new temporary data set that contains only observations from big states.

6. Print an alphabetical list of the state names and the populations of the big states. 

7. Estimate a regression using all of the states in which spending is the dependent variable and population and the big population state dummy are the independent variables.

Now run the program.

If there are any errors, do your best to try to debug.

Program for Exercise 1.

/* u:\rob\SAS workshop\ex1.sas performs the tasks in exercise 1*/

libname users 'c:\users\';

filename statetxt 'c:\users\states.txt';

/* 2. Bring in the ascii data and save as c:\users\stperm.sd2 */

data users.stperm;

  infile statetxt;

  input state $ 1-2 pop 12-19 inc_cp 20-25 spend 28-32;

/* create per capita spending*/

spend_cp = spend/pop;

/* 3. Display contents of the data file */

proc contents data = users.stperm;

/* 4. Means of per cap income and per cap spending */

PROC MEANS data = users.stperm;

   Var inc_cp spend_cp;

/* 5. Fsbrowse */

PROC FSBROWSE data = users.stperm;

/* 6. Create a new data set */

data nonyca (keep=state inc_cp income);

  set users.stperm;

/* drop NY and CA */

if state = 'NY' or state = 'CA' then delete;

income = inc_cp*pop;

/*keep state inc_cp income*/

run;

Program for Exercise 2.

/* u:\rob\SAS workshop\ex2.sas performs the tasks in exercise 2*/

libname users 'c:\users\';

/*1&2 - get data & create dummy*/

data sttemp;

  set users.stperm;

if pop >5000 then bigpop = 1;

  else bigpop = 0;

/* 3 freq distribution */

proc freq data = sttemp;

 tables bigpop;

/* 4 - means by bigpop*/

proc sort data = sttemp;

  by bigpop;

proc means data = sttemp;

  var spend spend_cp;

    by bigpop;

/*5.  new data set with only obs from big states*/

data bigstate;

  set sttemp;

 if bigpop = 0 then delete;

/* if bigpop =1;*/

/* 6. alphabetical list for bigstates*/

proc sort data = bigstate;

 by state;

proc print data = bigstate;

  var state pop;

/* 7. estimate spending = a+b1*pop+b2*bigpop */

proc reg data= sttemp;

 model spend = pop bigpop;

run;

